
Wallaby helps you test your web applications by simulating realistic user interactions.
By default it runs each test case concurrently and manages browsers for you.
Here's an example test for a simple Todo application:
defmodule MyApp.Features.TodoTest do
  use ExUnit.Case, async: true
  use Wallaby.Feature

  import Wallaby.Query, only: [css: 2, text_field: 1, button: 1]

  feature "users can create todos", %{session: session} do
    session
    |> visit("/todos")
    |> fill_in(text_field("New Todo"), with: "Write my first Wallaby test")
    |> click(button("Save"))
    |> assert_has(css(".alert", text: "You created a todo"))
    |> assert_has(css(".todo-list > .todo", text: "Write my first Wallaby test"))
  end
end
Because Wallaby manages multiple browsers for you, its possible to test several users interacting with a page simultaneously.
defmodule MyApp.Features.MultipleUsersTest do
  use ExUnit.Case, async: true
  use Wallaby.Feature

  import Wallaby.Query, only: [text_field: 1, button: 1, css: 2]

  @page message_path(Endpoint, :index)
  @message_field text_field("Share Message")
  @share_button button("Share")

  def message(msg), do: css(".messages > .message", text: msg)

  @sessions 2
  feature "That users can send messages to each other", %{sessions: [user1, user2]} do
    user1
    |> visit(@page)
    |> fill_in(@message_field, with: "Hello there!")
    |> click(@share_button)

    user2
    |> visit(@page)
    |> fill_in(@message_field, with: "Hello yourself")
    |> click(@share_button)

    user1
    |> assert_has(message("Hello yourself"))

    user2
    |> assert_has(message("Hello there!"))
  end
end
Read on to see what else Wallaby can do or check out the Official Documentation.

  
    
  
  Sponsors


Your company's name and logo could be here!

  
    
  
  Setup



  
    
  
  Requirements


Wallaby requires Elixir 1.12+ and OTP 22+.
Wallaby also requires bash to be installed. Generally bash is widely available, but it does not come pre-installed on Alpine Linux.

  
    
  
  Installation


Add Wallaby to your list of dependencies in mix.exs:
def deps do
  [
    {:wallaby, "~> 0.30", runtime: false, only: :test}
  ]
end
Configure the driver.
# Chrome
config :wallaby, driver: Wallaby.Chrome # default

# Selenium
config :wallaby, driver: Wallaby.Selenium
You'll need to install the actual drivers as well.
	Chrome
	chromedriver


	Selenium
	selenium
	geckodriver (for Firefox) or chromedriver (for Chrome)



Ensure that Wallaby is started in your test_helper.exs:
{:ok, _} = Application.ensure_all_started(:wallaby)
When calling use Wallaby.Feature and using Ecto, please configure your otp_app.
config :wallaby, otp_app: :your_app

  
    
  
  Phoenix


Enable Phoenix to serve endpoints in tests:
# config/test.exs

config :your_app, YourAppWeb.Endpoint,
  server: true
In your test_helper.exs you can provide some configuration to Wallaby.
At a minimum, you need to specify a :base_url, so Wallaby knows how to resolve relative paths.
# test/test_helper.exs

Application.put_env(:wallaby, :base_url, YourAppWeb.Endpoint.url)
Ecto
If you're testing a Phoenix application with Ecto and a database that supports sandbox mode, you can enable concurrent testing by adding the Phoenix.Ecto.SQL.Sandbox plug to your Endpoint.
It's important that this is at the top of endpoint.ex before any other plugs.
# lib/your_app_web/endpoint.ex

defmodule YourAppWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :your_app

  if sandbox = Application.compile_env(:your_app, :sandbox, false) do
    plug Phoenix.Ecto.SQL.Sandbox, sandbox: sandbox
  end

  # ...

  socket("/live", Phoenix.LiveView.Socket,
    websocket: [connect_info: [:user_agent, session: @session_options]]
  )
It's also important to make sure the user_agent is passed in the connect_info in order to allow the database and browser session to be wired up correctly.
Then make sure sandbox is enabled:
# config/test.exs

config :your_app, :sandbox, Ecto.Adapters.SQL.Sandbox
This enables the database connection to be owned by the process that is running your test, but the connection is shared to the process receiving the HTTP requests from the browser, so that the same data is visible in both processes.
If you have other resources that should be shared by both processes (e.g. expectations or stubs if using Mox), then you can define a custom sandbox module:
# test/support/sandbox.ex

defmodule YourApp.Sandbox do
  def allow(repo, owner_pid, child_pid) do
    # Delegate to the Ecto sandbox
    Ecto.Adapters.SQL.Sandbox.allow(repo, owner_pid, child_pid)

    # Add custom process-sharing configuration
    Mox.allow(MyMock, owner_pid, child_pid)
  end
end
And update the test config to use your custom sandbox:
# config/test.exs

config :your_app, :sandbox, YourApp.Sandbox
Assets
Assets are not re-compiled when you run mix test.
This can lead to confusion if you've made changes in JavaScript or CSS but tests are still failing.
There are two common ways to avoid this confusion.
esbuild
If you're using esbuild you can add esbuild default to the test alias in the mix config file.
  defp aliases do
    [
      "test": [
        "esbuild default",
        "ecto.create --quiet",
        "ecto.migrate",
        "test",
      ]
    ]
  end
Webpack
The first solution is to run webpack --mode development --watch from the assets directory.
This will ensure that assets get recompiled after any changes.
The second solution is to add a new alias to your mix config that recompiles assets for you:
  def project do
    [
      app: :my_app,
      version: "1.0.0",
      aliases: aliases()
    ]
  end

  defp aliases, do: [
    "test": [
      "assets.compile --quiet",
      "ecto.create --quiet",
      "ecto.migrate",
      "test",
    ],
    "assets.compile": &compile_assets/1
  ]

  defp compile_assets(_) do
    Mix.shell().cmd("cd assets && ./node_modules/.bin/webpack --mode development",
      quiet: true
    )
  end
This method is less error prone but it will cause a delay when starting your test suite.
LiveView
In order to test Phoenix LiveView (as of version 0.17.7) with Wallaby you'll also need to add a function to each mount/3 function in your LiveViews, or use the on_mount live_session lifecycle hook in the router:
defmodule MyApp.Hooks.AllowEctoSandbox do
  import Phoenix.LiveView
  import Phoenix.Component

  def on_mount(:default, _params, _session, socket) do
    allow_ecto_sandbox(socket)
    {:cont, socket}
  end

  defp allow_ecto_sandbox(socket) do
    %{assigns: %{phoenix_ecto_sandbox: metadata}} =
      assign_new(socket, :phoenix_ecto_sandbox, fn ->
        if connected?(socket), do: get_connect_info(socket, :user_agent)
      end)

    Phoenix.Ecto.SQL.Sandbox.allow(metadata, Application.get_env(:your_app, :sandbox))
  end
end
and then including the function usage in the router:
live_session :default, on_mount: MyApp.Hooks.AllowEctoSandbox do
  # ...
end
Umbrella Apps
If you're testing an umbrella application containing a Phoenix application for the web interface (MyWebApp) and a separate persistence application (MyPersistenceApp) using Ecto 2 or 3 with a database that supports sandbox mode, then you can use the same setup as above, with a few tweaks.
# my_web_app/lib/endpoint.ex

defmodule MyWebApp.Endpoint do
  use Phoenix.Endpoint, otp_app: :my_web_app

  if Application.get_env(:my_persistence_app, :sql_sandbox) do
    plug Phoenix.Ecto.SQL.Sandbox
  end
Make sure MyWebApp is set up to serve endpoints in tests and that the SQL sandbox is enabled:
# my_web_app/config/test.exs

config :my_web_app, MyWebApp.Endpoint,
  server: true

config :my_persistence_app, :sql_sandbox, true
Then in MyWebApp's test_helper.exs you can provide some configuration to Wallaby.
At minimum, you need to specify a :base_url, so Wallaby knows how to resolve relative paths.
# my_web_app/test/test_helper.exs

Application.put_env(:wallaby, :base_url, MyWebApp.Endpoint.url)
You will also want to add phoenix_ecto as a dependency to MyWebApp:
# my_web_app/mix.exs

def deps do
  [
    {:phoenix_ecto, "~> 3.0", only: :test}
  ]
end

  
    
  
  Writing tests


It's easiest to add Wallaby to your test suite by using the Wallaby.Feature module.
defmodule YourApp.UserListTest do
  use ExUnit.Case, async: true
  use Wallaby.Feature

  feature "users have names", %{session: session} do
    session
    |> visit("/users")
    |> find(Query.css(".user", count: 3))
    |> List.first()
    |> assert_has(Query.css(".user-name", text: "Chris"))
  end
end

  
    
  
  API


The full documentation for the DSL is in the official documentation.

  
    
  
  Queries and Actions


Wallaby's API is broken into 2 concepts: Queries and Actions.
Queries allow us to declaratively describe the elements that we would like to interact with and Actions allow us to use those queries to interact with the DOM.
Lets say that our html looks like this:
<ul class="users">
  <li class="user">
    <span class="user-name">Ada</span>
  </li>
  <li class="user">
    <span class="user-name">Grace</span>
  </li>
  <li class="user">
    <span class="user-name">Alan</span>
  </li>
</ul>
If we wanted to interact with all of the users then we could write a query like so css(".user", count: 3).
If we only wanted to interact with a specific user then we could write a query like this css(".user-name", count: 1, text: "Ada"). Now we can use those queries with some actions:
session
|> find(css(".user", count: 3))
|> List.first
|> assert_has(css(".user-name", count: 1, text: "Ada"))
There are several queries for common html elements defined in the Query module.
All actions accept a query.
This makes it easy to use queries we've already defined.
Actions will block until the query is either satisfied or the action times out.
Blocking reduces race conditions when elements are added or removed dynamically.

  
    
  
  Navigation


We can navigate directly to pages with visit:
visit(session, "/page.html")
visit(session, user_path(Endpoint, :index, 17))
It's also possible to click links directly:
click(session, link("Page 1"))

  
    
  
  Finding


We can find a specific element or list of elements with find:
@user_form   css(".user-form")
@name_field  text_field("Name")
@email_field text_field("Email")
@save_button button("Save")

find(page, @user_form, fn(form) ->
  form
  |> fill_in(@name_field, with: "Chris")
  |> fill_in(@email_field, with: "c@keathley.io")
  |> click(@save_button)
end)
Passing a callback to find will return the parent which makes it easy to chain find with other actions:
page
|> find(css(".users"), & assert has?(&1, css(".user", count: 3)))
|> click(link("Next Page"))
Without the callback find returns the element.
This provides a way to scope all future actions within an element.
page
|> find(css(".user-form"))
|> fill_in(text_field("Name"), with: "Chris")
|> fill_in(text_field("Email"), with: "c@keathley.io")
|> click(button("Save"))

  
    
  
  Interacting with forms


There are a few ways to interact with form elements on a page:
fill_in(session, text_field("First Name"), with: "Chris")
clear(session, text_field("last_name"))
click(session, option("Some option"))
click(session, radio_button("My Fancy Radio Button"))
click(session, button("Some Button"))
If you need to send specific keys to an element, you can do that with send_keys:
send_keys(session, ["Example", "Text", :enter])

  
    
  
  Assertions


Wallaby provides custom assertions to make writing tests easier:
assert_has(session, css(".signup-form"))
refute_has(session, css(".alert"))
has?(session, css(".user-edit-modal", visible: false))
assert_has and refute_has both take a parent element as their first argument.
They return that parent, making it easy to chain them together with other actions.
session
|> assert_has(css(".signup-form"))
|> fill_in(text_field("Email", with: "c@keathley.io"))
|> click(button("Sign up"))
|> refute_has(css(".error"))
|> assert_has(css(".alert", text: "Welcome!"))

  
    
  
  Window Size


You can set the default window size by passing in the window_size option into Wallaby.start_session\1.
Wallaby.start_session(window_size: [width: 1280, height: 720])
You can also resize the window and get the current window size during the test.
resize_window(session, 100, 100)
window_size(session)

  
    
  
  Screenshots


It's possible take screenshots:
take_screenshot(session)
All screenshots are saved to a screenshots directory in the directory that the tests were run in.
If you want to customize the screenshot directory you can pass it as a config value:
# config/test.exs
config :wallaby, screenshot_dir: "/file/path"

# test_helper.exs
Application.put_env(:wallaby, :screenshot_dir, "/file/path")

  
    
  
  Automatic screenshots


You can automatically take screenshots on an error when using the Wallaby.Feature.feature/3 macro.
# config/test.exs
config :wallaby, screenshot_on_failure: true

# test_helper.exs
Application.put_env(:wallaby, :screenshot_on_failure, true)

  
    
  
  JavaScript



  
    
  
  Asynchronous code


Testing asynchronous JavaScript code can expose timing issues and race conditions.
We might try to interact with an element that hasn't yet appeared on the page.
Elements can become stale while we're trying to interact with them.
Wallaby helps solve this by blocking.
Instead of manually setting timeouts we can use assert_has and some declarative queries to block until the DOM is in a good state.
session
|> click(button("Some Async Button"))
|> assert_has(css(".async-result"))
|> click(button("Next Action"))

  
    
  
  Interacting with dialogs


Wallaby provides several ways to interact with JavaScript dialogs such as window.alert, window.confirm and window.prompt.
You can use one of the following functions:
	For window.alert use accept_alert/2
	For window.confirm use accept_confirm/2 or dismiss_confirm/2
	For window.prompt use accept_prompt/2-3 or dismiss_prompt/2

All of these take a function as last parameter, which must include the necessary interactions to trigger the dialog. For example:
alert_message = accept_alert session, fn(session) ->
  click(session, link("Trigger alert"))
end
To emulate user input for a prompt, accept_prompt takes an optional parameter:
prompt_message = accept_prompt session, [with: "User input"], fn(session) ->
  click(session, link("Trigger prompt"))
end

  
    
  
  JavaScript logging and errors


Wallaby captures both JavaScript logs and errors.
Any uncaught exceptions in JavaScript will be re-thrown in Elixir.
This can be disabled by specifying js_errors: false in your Wallaby config.
JavaScript logs are written to :stdio by default.
This can be changed to any IO device by setting the :js_logger option in your Wallaby config.
For instance if you want to write all JavaScript console logs to a file you could do something like this:
{:ok, file} = File.open("browser_logs.log", [:write])
Application.put_env(:wallaby, :js_logger, file)
Logging can be disabled by setting :js_logger to nil.

  
    
  
  Configuration



  
    
  
  Adjusting timeouts


Wallaby uses hackney under the hood, so we offer a hook that allows you to control any hackney options you'd like to have sent along on every request.
This can be controlled with the :hackney_options setting in config.exs.
config :wallaby,
  hackney_options: [timeout: :infinity, recv_timeout: :infinity]

# Overriding a value
config :wallaby,
  hackney_options: [timeout: 5_000]

  
    
  
  Contributing


Wallaby is a community project. Pull Requests (PRs) and reporting issues are greatly welcome.
To get started and setup the project, make sure you've got Elixir 1.7+ installed and then:

  
    
  
  Development Dependencies


Wallaby requires the following tools.
	ChromeDriver
	Google Chrome
	GeckoDriver
	Mozilla Firefox
	selenium-server-standalone

# Unit tests
$ mix test

# Integration tests for all drivers
$ mix test.drivers

# Integration tests for a specific driver
$ WALLABY_DRIVER=chrome mix test
$ WALLABY_DRIVER=selenium mix test

# All tests
$ mix test.all



  

    
Wallaby 
    



      
A concurrent feature testing library.

  
    
  
  Configuration


Wallaby supports the following options:
	:otp_app - The name of your OTP application. This is used to check out your Ecto repos into the SQL Sandbox.
	:screenshot_dir - The directory to store screenshots.
	:screenshot_on_failure - if Wallaby should take screenshots on test failures (defaults to false).
	:max_wait_time - The amount of time that Wallaby should wait to find an element on the page. (defaults to 3_000)
	:js_errors - if Wallaby should re-throw JavaScript errors in elixir (defaults to true).
	:js_logger - IO device where JavaScript console logs are written to. Defaults to :stdio. This option can also be set to a file or any other io device. You can disable JavaScript console logging by setting this to nil.

    Wallaby.Browser - wallaby v0.30.9
    
    

    


  
  

    
Wallaby.Browser 
    



      
The Browser module is the entrypoint for interacting with a real browser.
By default, action only work with elements that are visible to a real user.

  
    
  
  Actions


Actions are used to interact with form elements. All actions work with the
query interface:
<label for="first_name">
  First Name
</label>
<input id="user_first_name" type="text" name="first_name">
fill_in(page, Query.text_field("First Name"), with: "Grace")
fill_in(page, Query.text_field("first_name"), with: "Grace")
fill_in(page, Query.text_field("user_first_name"), with: "Grace")
These queries work with any of the available actions.
fill_in(page, Query.text_field("First Name"), with: "Chris")
clear(page, Query.text_field("user_email"))
click(page, Query.radio_button("Radio Button 1"))
click(page, Query.checkbox("Checkbox"))
click(page, Query.checkbox("Checkbox"))
click(page, Query.option("Option 1"))
click(page, Query.button("Some Button"))
attach_file(page, Query.file_field("Avatar"), path: "test/fixtures/avatar.jpg")
Actions return their parent element so that they can be chained together:
page
|> find(Query.css(".signup-form"), fn(form) ->
  form
  |> fill_in(Query.text_field("Name"), with: "Grace Hopper")
  |> fill_in(Query.text_field("Email"), with: "grace@hopper.com")
  |> click(Query.button("Submit"))
end)

  
    
  
  Scoping


Finders provide scoping like so:
session
|> visit("/page.html")
|> find(Query.css(".users"))
|> find(Query.css(".user", count: 3))
|> List.first
|> find(Query.css(".user-name"))
If a callback is passed to find then the scoping will only apply to the callback
and the parent will be passed to the next action in the chain:
page
|> find(Query.css(".todo-form"), fn(form) ->
  form
  |> fill_in(Query.text_field("What needs doing?"), with: "Write Wallaby Documentation")
  |> click(Query.button("Save"))
end)
|> find(Query.css(".success-notification"), fn(notification) ->
  assert notification
  |> has_text?("Todo created successfully!")
end)
This allows you to create a test that is logically grouped together in a single pipeline.
It also means that its easy to create re-usable helper functions without having to worry about
chaining. You could re-write the above example like this:
def create_todo(page, todo) do
  find(Query.css(".todo-form"), & fill_in_and_save_todo(&1, todo))
end

def fill_in_and_save_todo(form, todo) do
  form
  |> fill_in(Query.text_field("What needs doing?"), with: todo)
  |> click(Query.button("Save"))
end

def todo_was_created?(page) do
  find Query.css(page, ".success-notification"), fn(notification) ->
    assert notification
    |> has_text?("Todo created successfully!")
  end
end

assert page
|> create_todo("Write Wallaby Documentation")
|> todo_was_created?

Wallaby.Chrome 
    



      
The Chrome driver uses Chromedriver to power Google Chrome and Chromium.

  
    
  
  Usage


Start a Wallaby Session using this driver with the following command:
{:ok, session} = Wallaby.start_session()

  
    
  
  Configuration



  
    
  
  Headless


Chrome will run in headless mode by default.
You can disable this behaviour using the following configuration.
This will override the default capabilities and capabilities set with application configuration.
This will not override capabilities passed in directly to Wallaby.start_session/1.
config :wallaby,
  chromedriver: [
    headless: false
  ]

  
    
  
  Capabilities


These capabilities will override the default capabilities.
config :wallaby,
  chromedriver: [
    capabilities: %{
      # something
    }
  ]

  
    
  
  ChromeDriver binary


If ChromeDriver is not available in your path, you can specify it's location.
config :wallaby,
  chromedriver: [
    path: "path/to/chrome"
  ]

  
    
  
  Chrome binary


This configures which instance of Google Chrome to use.
This will override the default capabilities and capabilities set with application configuration.
This will not override capabilities passed in directly to Wallaby.start_session/1.
config :wallaby,
  chromedriver: [
    binary: "path/to/chrome"
  ]

  
    
  
  Default Capabilities


By default, Chromdriver will use the following capabilities
You can read more about capabilities in the JSON Wire Protocol documentation and the Chromedriver documentation.
  %{
    javascriptEnabled: false,
    loadImages: false,
    version: "",
    rotatable: false,
    takesScreenshot: true,
    cssSelectorsEnabled: true,
    nativeEvents: false,
    platform: "ANY",
    unhandledPromptBehavior: "accept",
    loggingPrefs: %{
      browser: "DEBUG"
    },
    chromeOptions: %{
      args: [
        "--no-sandbox",
        "window-size=1280,800",
        "--disable-gpu",
        "--headless",
        "--fullscreen",
        "--user-agent=Mozilla/5.0 (Windows NT 6.1) AppleWebKit/537.36 (KHTML, like Gecko) Chrome/41.0.2228.0 Safari/537.36"
      ]
    }
  }

  
    
  
  Notes


This driver requires Chromedriver to be installed in your path.

Sets up the Wallaby DSL in a module.
All functions in Wallaby.Browser are now accessible without a module name
and Wallaby.Browser, Wallaby.Element and Wallaby.Query are all aliased.

  
    
  
  Example


defmodule MyPage do
  use Wallaby.DSL

  @name_field Query.text_field("Name")
  @email_field Query.text_field("email")
  @save_button Query.button("Save")

  def register(session) do
    session
    |> visit("/registration.html")
    |> fill_in(@name_field, with: "Chris")
    |> fill_in(@email_field, with: "c@keathly.io")
    |> click(@save_button)
  end
end

Wallaby.Element 
    



      
Defines an Element Struct and interactions with Elements.
Typically these functions are used in conjunction with a find:
page
|> find(Query.css(".some-element"), fn(element) -> Element.click(element) end)
These functions can be used to create new actions specific to your application:
def create_todo(todo_field, todo_text) do
  todo_field
  |> Element.click()
  |> Element.fill_in(with: todo_text)
  |> Element.send_keys([:enter])
end

  
    
  
  Retrying


Unlike Browser the actions in Element do not retry if the element becomes stale. Instead an exception will be raised.

Helpers for writing features.
You can use or import this module.

  
    
  
  use Wallaby.Feature


Calling this module with use will automatically call use Wallaby.DSL.
When called with use and you are using Ecto, please configure your otp_app.
config :wallaby, otp_app: :your_app

Shortcuts for various keys.
	:null
	:cancel
	:help
	:backspace
	:tab
	:clear
	:return
	:enter
	:shift
	:control
	:alt
	:pause
	:escape
	:space
	:pageup
	:pagedown
	:end
	:home
	:left_arrow
	:up_arrow
	:right_arrow
	:down_arrow
	:insert
	:delete
	:semicolon
	:equals
	:num0
	:num1
	:num2
	:num3
	:num4
	:num5
	:num6
	:num7
	:num8
	:num9
	:multiply
	:add
	:seperator
	:subtract
	:decimal
	:divide
	:command

Provides the query DSL.
Queries are used to locate and retrieve DOM elements from a browser (see
Wallaby.Browser). You create queries like so:
Query.css(".some-css")
Query.xpath(".//input")

  
    
  
  Form elements


There are several custom finders for locating form elements. Each of these allows
finding by their name, id text, or label text. This allows for more robust querying
and decouples the query from presentation selectors like CSS classes.
Query.text_field("My Name")
Query.checkbox("Checkbox")
Query.select("A Fancy Select Box")

  
    
  
  Query Options


All of the query operations accept the following options:
	:count - The number of elements that should be found or :any (default: 1).	If a :minimum or :maximum is specified, it defaults to nil.


	:minimum - The minimum number of elements that should be found, or nil (default: nil).
	:maximum - The maximum number of elements that should be found, or nil (default: nil).
	:visible - Determines if the query should return only visible elements (default: true).
	:selected - Determines if the query should return only selected elements (default: :any).
	:text - Text that should be found inside the element (default: nil).
	:at - The position (a number or :all) of the element to return if multiple elements satisfy the query. (default: :all)

Query options can also be set via functions by the same names:
Query.css(".names")
|> Query.visible(true)
|> Query.count(3)

  
    
  
  Re-using queries


It is often convenient to re-use queries. The easiest way is to use module
attributes:
@name_field Query.text_field("User Name")
@submit_button Query.button("Save")
If the queries need to be dynamic then you should create a module that
encapsulates the queries as functions:
defmodule TodoListPage do
  def todo_list do
    Query.css(".todo-list")
  end

  def todos(count) do
    Query.css(".todo", count: count)
  end
end

  
    
  
  What does my query do?


Wanna check out what exactly your query will do? Look no further than
Wallaby.Query.compile/1 - it takes a query and returns the CSS or xpath
query that will be sent to the driver:
iex> Wallaby.Query.compile Wallaby.Query.text("my text")
{:xpath, ".//*[contains(normalize-space(text()), \"my text\")]"}
So, whenever you're not sure whatever a specific query will do just compile
it to get all the details!

Wallaby.Selenium 
    



      
The Selenium driver uses Selenium Server to power many types of browsers (Chrome, Firefox, Edge, etc).

  
    
  
  Usage


Start a Wallaby Session using this driver with the following command:
{:ok, session} = Wallaby.start_session()

  
    
  
  Configuration



  
    
  
  Capabilities


These capabilities will override the default capabilities.
config :wallaby,
  selenium: [
    capabilities: %{
      # something
    }
  ]

  
    
  
  Selenium Remote URL


It is possible to globally set Selenium's "Remote URL" by setting the following option.
By default it is http://localhost:4444/wd/hub/
config :wallaby,
  selenium: [
    remote_url: "http://selenium_url"
  ]

  
    
  
  Default Capabilities


By default, Selenium will use the following capabilities
You can read more about capabilities in the JSON Wire Protocol documentation.
%{
  javascriptEnabled: true,
  browserName: "firefox",
  "moz:firefoxOptions": %{
    args: ["-headless"]
  }
}

  
    
  
  Notes


	Requires selenium-server to be running on port 4444. Wallaby does not manage the start/stop of the Selenium server.
	Requires GeckoDriver to be installed in your path when using Firefox. Firefox is used by default.

Wallaby.Session 
    



      
Struct containing details about the webdriver session.

Wallaby.BadMetadataError exception

Wallaby.ExpectationNotMetError exception
    






  

  
    
    Wallaby.JSError - wallaby v0.30.9
    
    

    


  
  

    
Wallaby.JSError exception

